
1
Introduction

There is a major issue with the MIKEY type definitions in TTCN: Different codec implementations may need templates to be structured in different ways for one and the same MIKEY message. 
( MIKEY type definitions need to be re-structured.
2
Proposal

New type definitions to be used for MCX test case implementation fulfilling the following requirements:
· Unambiguous encoding/decoding

· Support of MIKEY SAKKE mode and all payloads being required 

· Potential extendibility for other flavours of MIKEY
· Generic handling of “Next Payload” fields
Implementation
	module MIKEY_TypeDefs {

  import from CommonDefs all;

  //============================================================================

  // MIKEY message - RFC 3830 clause 6 with extensions in RFC 6509 and RFC 6043

  //----------------------------------------------------------------------------

  type record MIKEY_Message  {

    MIKEY_CommonHeader          commonHdr,

    MIKEY_PayloadList           payloads,

    MIKEY_Signature_Payload     sign  optional

  };

  //============================================================================

  // Common Header payload (HDR)

  //----------------------------------------------------------------------------

  type record MIKEY_CommonHeader  {             /* RFC 3830 clause 6.1: Common Header payload (HDR) */

    O1_Type version,

    O1_Type dataType,

    O1_Type nextPayload,

    B1_Type v,

    B7_Type prfFunc,

    O4_Type csbId,

    O1_Type numberCS,

    O1_Type csIdMap,                            /* CS ID map type:

                                                   '00'O   SRTP-ID (RFC 3830 clause 6.1)

                                                   '02'O   GENERIC-ID (RFC 6043 clause 6.1)*/

    MIKEY_CryptoSession csIdMapInfo             /* CS ID map info depending on CS ID map type */

  };

  type union MIKEY_CryptoSession {              /* crypto session which correspond to the CS ID map value in the Common Header payload (HDR) */

    Null_Type                   SRTP,           /* place holder for SRTP ID according to RFC 3830 clause 6.1.1 */

    MIKEY_GenericIdMapType      GENERIC         /* GENERIC-ID Map Type according to RFC 6043 clause 6.1.1 */

  };

  type record MIKEY_GenericIdMapType  {         /* GENERIC-ID Map Type according to RFC 6043 clause 6.1.1 */

    O1_Type                     csId,

    O1_Type                     protType,

    B1_Type                     s,

    B7_Type                     nP,

    MIKEY_Ps                    ps,

    O2_Type                     sessionDataLen,

    octetstring                 sessionData,

    O1_Type                     spiLen,

    octetstring                 spi

  };

  type set of O1_Type MIKEY_Ps; // List of policies RFC 6043 clause 6.1.1

  //============================================================================

  // Other payloads

  //----------------------------------------------------------------------------

  type record MIKEY_GenericPayload {

    O1_Type                     nextPayload,

    MIKEY_PayloadContent        content

  };

  type set of MIKEY_GenericPayload MIKEY_PayloadList;

  type union MIKEY_PayloadContent {                             // NOTE: further payloads may need to be added, when MIKEY messages other than MIKEY SAKKE shall be used 

    MIKEY_Timestamp_Payload             Timestamp,              // RFC 3830 clause 6.6:  Timestamp Payload (T)

    MIKEY_SecurityPolicy_Payload        SecurityPolicy,         // RFC 3830 clause 6.10: Security Policy Payload (SP)

    MIKEY_RAND_Payload                  RAND,                   // RFC 3830 clause 6.11: RAND Payload (RAND)

    MIKEY_GeneralExtension_Payload      Extension,              // RFC 3830 clause 6.15: General Extension Payload

    MIKEY_IDR_Payload                   IDR,                    // RFC 6043 clause 6.6:  ID Payload with Role Indicator (IDR)

    MIKEY_SAKKE_Payload                 SAKKE                   // RFC 6509 clause 4.2:  SAKKE Payload

  };

  type record MIKEY_Signature_Payload  {                        // RFC 3830 clause 6.5:  Signature Payload (SIGN)

    B4_Type sType,

    B12_Type sLen,

    octetstring s

  };

  type record MIKEY_Timestamp_Payload  {                        // RFC 3830 clause 6.6:  Timestamp Payload (T)

    O1_Type tsType,

    octetstring tsValue

  };

  type record MIKEY_SecurityPolicy_Payload {                    // RFC 3830 clause 6.10: Security Policy Payload (SP)

    O1_Type policyNo,

    O1_Type protType,

    O2_Type policyParamLen,

    MIKEY_PolicyParams policyParams 

  };

  type set of MIKEY_PolicyParam  MIKEY_PolicyParams;

  type record MIKEY_PolicyParam  {

    O1_Type policyType,

    O1_Type len,

    octetstring val

  };

  type record MIKEY_RAND_Payload  {                             // RFC 3830 clause 6.11: RAND Payload (RAND)

    O1_Type randLen,

    octetstring randValue

  };

  type record MIKEY_GeneralExtension_Payload  {                 // RFC 3830 clause 6.15: General Extension Payload

    O1_Type extType,

    O2_Type extLen,

    octetstring extData

  };

  type record MIKEY_IDR_Payload  {                              // RFC 6043 clause 6.6:  ID Payload with Role Indicator (IDR)

    O1_Type idRole,

    O1_Type idType,

    O2_Type idLen,

    octetstring idData 

  };

  type record MIKEY_SAKKE_Payload  {                            // RFC 6509 clause 4.2:  SAKKE Payload

    O1_Type sakkeParms,

    O1_Type idScheme,

    O2_Type sakkeLen,

    octetstring sakkeData 

  };

} with {encode "MIKEY-Codec"}

